Vitória Menino Campos 10874175

Agenor de Toledo Fleury

PME 3380 - Modelagem de Sistemas Dinâmicos

01 de outubro de 2020

LISTA D

EXEMPLO:

A solução numérica da equação diferencial ordinária, linear, a parâmetros constantes pode ser determinada no Scilab. Um sistema linear pode ser representado por:

$$\dot{\mathbf{x}} = \mathbf{Ax} + \mathbf{Bu} \qquad \text{(equações diferenciais – equações de estado)}$$

$$\mathbf{y} = \mathbf{Cx} + \mathbf{Du} \qquad \text{(equações algébricas – equações da saída)}$$

No caso do exemplo do reservatório:

$$\dot{x} = \underbrace{-\frac{1}{2S}\sqrt{\frac{\rho g}{Rh_o}}}_{A}x + \underbrace{\frac{1}{S}}_{B}u \qquad \text{(equações diferenciais)}$$

$$y = \underbrace{+1}_{C}x + \underbrace{0}_{D}u \qquad \text{(equações algébricas)}$$

O gráfico gerado a partir do código de exemplo é:



<u>EXERCÍCIO</u>:

<u>PARTE 1:</u> Faça as modificações adequadas para se poder desenhar e comparar os gráficos da resposta do sistema não linear e linear. Faça as simulações dos sistemas linear e não linear considerando que o reservatório parte do nível **h = 2 m**, mas com **vazão de entrada nula**. Compare as respostas.

O gráfico obtido a partir das condições iniciais, ho=2 e Qei=0, está representado abaixo:



Podemos observar que a curva não-linear tem resposta mais rápida, comparando com a curva linear, que tem perfil exponencial inverso.

PARTE 2: Obtenha o modelo matemático do circuito elétrico mostrado abaixo e compare com o modelo linear do sistema com um reservatório. Faça simulações e compare qualitativamente com os resultados do exercício 1 (sistema linear).



Utilizando as Leis de kirchhoff para encontrar a equação diferencial:

$$e_i = R.i + e_o \qquad onde \qquad e_0 = \frac{q}{C} \quad e \quad i = \frac{q}{t} = \widehat{q}$$

Substituindo na e_o e i na equação de e_i e isolando $\widehat{q}$:

$$\widehat{q} = -\frac{1}{RC}.q + \frac{e_i}{R}$$

Temos uma equação na forma $\widehat{x} = Ax + B$, semelhante a do modelo de um reservatório, o gráfico que modela o comportamento deste circuito é o seguinte:



LIÇÃO DE CASA:

PARTE 1: Usando a abordagem vista nestes exemplos, faça a simulação do sistema com dois reservatórios, supondo o modelo linear:



Da lista C temos as equações que modelam o sistema com dois reservatórios:

$$\dot{h}_1 = [Q_e - \sqrt{\frac{\rho g}{R_a}(h_1 - h_2)} - \sqrt{\frac{\rho g}{R_s} h_2}]\ \frac{1}{S_1}$$

$$\dot{h}_2 = [\sqrt{\frac{\rho g}{R_a}(h_1 - h_2)} - \sqrt{\frac{\rho g}{R_s} h_2}]\ \frac{1}{S_2}$$

E também seu modelo linearizado:

$$f_1(h_1, h_2, Q_e) = -\frac{1}{2S_1\sqrt{\frac{Ra}{\rho g}(h_{1_0} - h_{2_2})}}(h_1 + h_{1_0}) + \frac{1}{2S_1\sqrt{\frac{Ra}{\rho g}(h_{1_0} - h_{2_2})}}(h_2 - h_{2_0}) + \frac{1}{S_1}(Q_e - Q_{e_0})$$

$$f_2(h_1, h_2, Q_e) = \frac{1}{2S_2\sqrt{\frac{Ra}{\rho g}(h_{1_0} - h_{2_2})}}(h_1 + h_{1_0}) - \frac{1}{2S_2}[\frac{1}{\sqrt{\frac{Ra}{\rho g}(h_{1_0} - h_{2_0})}} + \frac{1}{\sqrt{\frac{Ra}{\rho g}(h_{2_0})}}](h_2 - h_{2_0})$$

Que pode ser escrito como:

$$\begin{bmatrix} \dot{x}_1 \\ \dot{x}_2 \end{bmatrix} = \begin{bmatrix} \alpha_1 & \alpha_2 \\ \alpha_3 & \alpha_4 \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \end{bmatrix} + \begin{bmatrix} \beta \\ 0 \end{bmatrix} u$$

$$\begin{bmatrix} y_1 \\ y_2 \end{bmatrix} = \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \end{bmatrix}$$

O gráfico obtido a partir das condições iniciais, $h_{o1}=2$, $h_{02}=4$ e vazão de entrada nula, está representado abaixo:



<u>PARTE 2:</u> Desenvolva um circuito elétrico análogo ao sistema com dois reservatórios.



CÓDIGOS:

Exercício - Parte 1:

```
// Eh sempre melhor apagar as variaveis anteriores
clear all
S=10; // [m^2] Area da secao transversal do reservatorio
rho=1000; // [kg/m^3] massa especifica da agua
g=10; // [m/s^2] aceleração da gravidade na superficie da Terra
R=2*10^8; // [Pa/(m^3/s)^2] parametro que relaciona pressao e vazao
ho=2; // [m] nivel do reservatorio em regime
hi=0.1; // [m] nivel adicional desejado
Qei=0; // [m^3/s] vazao na entrada


// Definir o sistema linear usando o comando syslin:
A=(-1/(2*S))*sqrt(rho*g/(R*ho));
B=1/S;
C=1;
D=0;
tanque=syslin('c',A,B,C,D); // o parametro 'c' indica que o sistema eh
 // continuo no tempo
// Definir a condicao inicial:
x0=2; // [m] desvio inicial do nivel em relação ao equilibrio
// Definir o vetor de instantes de tempo:
t=0:10:40000;
// Definir o vetor de entradas:
u=Qei*ones(t);
// Simulando o sistema usando o comando csim:
[y,x]=csim(u,t,tanque,x0);


//Definir o sistema nao linear:
function [hponto]=TNL(t, h, Qe)
    hponto = (Qe(t)-sqrt(rho*g*h/R))/S
endfunction

function [u]=vazao(t)
    u=0.000001;
endfunction

h = ode(ho,t(1),t,list(TNL,vazao));

// Plotando o resultado:
plot2d(t,y,2)
plot2d(t,h,6)
hl=legend(['Linear';'Não linear']);
// Colocando um titulo na figura e nomeando os eixos:
xtitle("Nível do tanque em respeito ao tempo","Tempo t [s]","Nivel [m]");
// Colocando uma grade azul no grafico:
xgrid(2)
```

Exercício - Parte 2:

```
clear all
xdel()
eo = 3;
R=100*10^3;
C=10*10^-6;
t=0:0.001:5;
function [qponto]=crte(t)
 qponto = eo/R*exp(-t/(R*C));
endfunction
i=crte(t)
plot2d(t,i,5)
xtitle("Circuito Elétrico","Tempo t [s]","Corrente [A]");
xgrid(2)
```

Lição de casa - Parte 1:

```
clear all
xdel()
S=10; // [m^2] Area da secao transversal do reservatorio
rho=1000; // [kg/m^3] massa especifica da agua
g=10; // [m/s^2] aceleração da gravidade na superficie da Terra
R=2*10^8; // [Pa/(m^3/s)^2] parametro que relaciona pressao e vazao
ho=2; // [m] nivel do reservatorio em regime
hi=0.1; // [m] nivel adicional desejado
Qeo=(1/2)*sqrt(rho*g/(R*ho))*hi; // [m^3/s] vazao na entrada
c=(rho*g)/(2*S*Qeo*R);
A = [-c,c;c,-2*c];
B = [1/S;0];
C = [1,0;0,1];
D = [0;0];
tanque=syslin('c',A,B,C,D);
h01=2;
h02=4;
hini = [h01;h02];
t=0:10:1000;
u=Qeo*ones(t);
[y,x]=csim(u,t,tanque,hini);
h1 = x(1,:);
h2 = x(2,:);
plot2d(t,h1,5)
plot2d(t,h2,3)
hl=legend(['Tanque 1';'Tanque 2']);
xtitle("Resposta de dois tanques: Sistema Linear","Tempo t [s]","Altura do nível [m]");
xgrid(2)
```